


Coupled Structural/Thermal Analysis

Introduction

This tutorial was completed using ANSYS 7.0. The purpose of this tutorial is to outline a simple coupled thermal/structural analysis. A steel link, with no internal stresses, is pinned between two solid structures at a reference temperature of 0 C (273 K). One of the solid structures is heated to a temperature of 75 C (348 K). As heat is transferred from the solid structure into the link, the link will attempt to expand. However, since it is pinned this cannot occur and as such, stress is created in the link. A steady-state solution of the resulting stress will be found to simplify the analysis.

Loads will not be applied to the link, only a temperature change of 75 degrees Celsius. The link is steel with a modulus of elasticity of 200 GPa, a thermal conductivity of 60.5 W/m*K and a thermal expansion coefficient of 12×10^{-6} /K.



Preprocessing: Defining the Problem

According to Chapter 2 of the ANSYS Coupled-Field Guide, "A sequentially coupled physics analysis is the combination of analyses from different engineering disciplines which interact to solve a global engineering problem. For convenience, ...the solutions and procedures associated with a particular engineering discipline [will be referred to as] a physics analysis. When the input of one physics analysis depends on the results from another analysis, the analyses are coupled."

Thus, each different physics environment must be constructed separately so they can be used to determine the coupled physics solution. However, it is important to note that a single set of nodes will exist for the entire model. By creating the geometry in the first physical environment, and using it with any following coupled environments, the geometry is kept constant. For our case, we will create the geometry in the Thermal Environment, where the thermal effects will be applied.

Although the geometry must remain constant, the element types can change. For instance, thermal elements are required for a thermal analysis while structural elements are required to determine the stress in the link. It is important to note, however that only certain combinations of elements can be used for a coupled physics analysis. For a listing, see Chapter 2 of the ANSYS Coupled-Field Guide located in the help file.

The process requires the user to create all the necessary environments, which are basically the preprocessing portions for each environment, and write them to memory. Then in the solution phase they can be combined to solve the coupled analysis.

ANSYS Command Listing

```

finish
/clear

/title, Thermal Stress Example
/prep7                                ! Enter preprocessor

k,1,0,0                                ! Keypoints
k,2,1,0

l,1,2                                    ! Line connecting keypoints

et,1,link33                             ! Element type
r,1,4e-4,                                ! Area
mp,kxx,1,60.5                            ! Thermal conductivity

esize,0.1                               ! Element size
lmesh,all                                ! Mesh line
physics,write,thermal                    ! Write physics environment as thermal
physics,clear                             ! Clear the environment

etchg,tts                                ! Element type
mp,ex,1,200e9                            ! Young's modulus
mp,prxy,1,0.3                            ! Poisson's ratio
mp,alpx,1,12e-6                          ! Expansion coefficient
physics,write,struct                      ! Write physics environment as struct
physics,clear
finish

/solu                                    ! Enter the solution phase
antype,0                                  ! Static analysis
physics,read,thermal                      ! Read in the thermal environment
dk,1,temp,348                             ! Apply a temp of 75 to keypoint 1

solve
finish

/solu                                    ! Re-enter the solution phase
physics,read,struct                       ! Read in the struct environment

ldread,temp,,,,,,,,rth                    ! Apply loads derived from thermal environment
tref,273

dk,1,all,0                                ! Apply structural constraints
dk,2,UX,0

```

```
solve  
finish
```

```
/post1                ! Enter postprocessor  
etable,CompStress,LS,1  ! Create an element table for link stress  
PRETAB,CompStress      ! Print the element table
```